	FACTOR ANALYSIS FOR NOMINAL DATA  	21




Supplementary materials to 

[bookmark: OLE_LINK35][bookmark: OLE_LINK36]Factor analysis for nominal (first choice) data









Appendix A: Assertiveness situational judgement test
1. You have spent ￥100 to buy a ticket to attend a musical concert. Security at the concert’s entrance refuses to let you in, claiming that the ticket you hold is a fake. It is a genuine ticket. You…
A. Ask to talk to the supervisor
B. Call your parents for suggestions
C. After arguing with the concert’s security, you give up and leave
 
2. At the end of a sports event (football or basketball match), 1.5 miles from home, you realize that you transportation plan back home has failed. You….
A. Call your parents for a ride
B. Don’t see any friends while leaving, but you see some acquaintances and ask them for a ride home
C. Walk back home
 
3. Imagine you work at a convenience store and your boss asks you to do 2 hours overtime. If you accept, you will one hour late to dinner at a restaurant with friends. You….
A. Call your friends letting them know you won’t come to dinner
B. Call your friends to let them know you’ll come after dinner
C. Let your boss know that you cannot do overtime that evening.
 
4. You and your friends have planned on staying out until X hours on a special night (graduation, New Year’s eve, etc.) but your parents ask you to be back at home at X – 2 hours. You…
A. Come back at the time set by your parents.
B. Try to convince your parents to let you be out for as long as your friends
C. Simply come back home when your friends do.
 
5. Your parents insist that you select a major which you are not interested in. You…
A. Select the major without complaining
B. Try to persuade your parents that the major is not for you
C. Talk to your relatives so that they help you convince your parents not to select it
 
6. You are going out for dinner with friends but you are yet to agree on a restaurant. You
A. Suggest a restaurant and push your friends to go there.
B. Listen to what your friends propose.
C. Suggest a restaurant, but don’t insist on going there.
 
7. In a school essay, you received a lower grade than you expected. You…
A. Talk to the instructor to complain about your grade
B. Do nothing; the instructor hates you, there is no point in complaining about your grade
C. Resolve to work harder next time to get a higher grade
 
8. Your friends and you are putting together money for a present to one of you. Your friends propose a larger amount per person than you can afford (or you were thinking). You…
A. Go along with the group
B. Try to convince them that the amount per person is excessive
C. Tell them how much money you can put at this time.

[bookmark: OLE_LINK6]Appendix B: Mplus code for the assertiveness application
The Mplus code to estimate this model is:
TITLE: Factor analysis of assertiveness data;
DATA:
    FILE IS assertiveness.dat;
          
VARIABLE:
    NAMES ARE i1-i8 age male;
    USEVARIABLES ARE i1-i8;
!This sets the items to be nominal
    NOMINAL ARE i1-i8;

ANALYSIS:
    ESTIMATOR = ML;

MODEL:
   assertive BY i1#1*1; ! we provide starting values so that the factor
   assertive BY i1#2*1; ! being measured be assertiveness

   assertive BY i2#1*1;
   assertive BY i2#2*1;

   assertive BY i3#1*-1;
   assertive BY i3#2*1;

   assertive BY i4#1*1;
   assertive BY i4#2*1;

   assertive BY i5#1*-1;
   assertive BY i5#2*1;

   assertive BY i6#1*1;
   assertive BY i6#2*-1;

   assertive BY i7#1*-1;
   assertive BY i7#2*-1;

   assertive BY i8#1*-1;
   assertive BY i8#2*1;
 
  assertive@1;   !Factor variance is set at 1.

OUTPUT: SVALUES;

SAVEDATA: FILE IS assertiveness fscores.dat;
    SAVE IS FSCORES;

[bookmark: _GoBack]The SVALUES command provides a list of all the model parameters with starting values equal to the estimated parameters. Replacing * by @ in these commands one obtains a file that can be used to score new observations on this questionnaire. 
TITLE: Scoring new observations in the assertiveness situational questionnaire;
DATA:
    FILE IS additional data.dat;
          
VARIABLE:
    NAMES ARE i1-i8;    NOMINAL ARE i1-i8;

MODEL:
     assertive BY i1#1@0.75254;
     assertive BY i1#2@0.81457;
     assertive BY i2#1@0.12445;
     assertive BY i2#2@0.18447;
     assertive BY i3#1@-0.29508;
     assertive BY i3#2@0.38766;
     assertive BY i4#1@0.33765;
     assertive BY i4#2@1.14062;
     assertive BY i5#1@-1.11344;
     assertive BY i5#2@0.20073;
     assertive BY i6#1@0.03573;
     assertive BY i6#2@-1.02894;
     assertive BY i7#1@-0.51764;
     assertive BY i7#2@-1.23319;
     assertive BY i8#1@-0.60544;
     assertive BY i8#2@0.34151;

     [ i1#1@2.52548 ];
     [ i1#2@-0.04635 ];
     [ i2#1@-0.51626 ];
     [ i2#2@-1.07375 ];
     [ i3#1@-0.40945 ];
     [ i3#2@0.89763 ];
     [ i4#1@0.67082 ];
     [ i4#2@3.81739 ];
     [ i5#1@-1.09640 ];
     [ i5#2@2.94000 ];
     [ i6#1@-3.52371 ];
     [ i6#2@-2.24498 ];
     [ i7#1@-2.35553 ];
     [ i7#2@-3.20627 ];
     [ i8#1@0.20486 ];
     [ i8#2@0.40931 ];

     assertive@1;

SAVEDATA: FILE IS new scores.dat;
    SAVE IS FSCORES;
   
[bookmark: OLE_LINK7]
Appendix C: Assertiveness application parameterized with deviation constraints in Mplus and flexMIRT

The deviation constraints parameterization is estimated by defining new parameters using the MODEL CONSTRAINT command of Mplus. These new parameters are labeled with letters v (the slopes/factor loadings) and t (the intercepts/thresholds) in the code below, and are subject to deviation constraints.

TITLE: Factor analysis of assertiveness data;
DATA:
    FILE IS assertiveness.dat;
         
VARIABLE:
    NAMES ARE i1-i8;
    USEVARIABLES ARE i1-i8;
!This sets the items to be nominal
    NOMINAL ARE i1-i8;

ANALYSIS:
    ESTIMATOR = ML;

MODEL:
   ! it is necessary to label each intercept and slope
   assertive BY i1#1*1  (b11);
   assertive BY i1#2*1  (b12);
   assertive BY i2#1*1  (b21);
   assertive BY i2#2*1  (b22);
   assertive BY i3#1*-1 (b31);
   assertive BY i3#2*1  (b32);
   assertive BY i4#1*1  (b41);
   assertive BY i4#2*1  (b42);
   assertive BY i5#1*-1 (b51);
   assertive BY i5#2*1  (b52);
   assertive BY i6#1*1  (b61);
   assertive BY i6#2*-1 (b62);
   assertive BY i7#1*-1 (b71);
   assertive BY i7#2*-1 (b72);
   assertive BY i8#1*-1 (b81);
   assertive BY i8#2*1  (b82);
   [i1#1] (a11);
   [i1#2] (a12);
   [i2#1] (a21);
   [i2#2] (a22);
   [i3#1] (a31);
   [i3#2] (a32);
   [i4#1] (a41);
   [i4#2] (a42);
   [i5#1] (a51);
   [i5#2] (a52);
   [i6#1] (a61);
   [i6#2] (a62);
   [i7#1] (a71);
   [i7#2] (a72);
   [i8#1] (a81);
   [i8#2] (a82);
      
   !Factor variance is set at 1.
    assertive@1;

MODEL CONSTRAINT: 
NEW (v11 v12 v13 v21 v22 v23 v31 v32 v33 v41 v42 v43 
v51 v52 v53 v61 v62 v63 v71 v72 v73 v81 v82 v83
t11 t12 t13 t21 t22 t23 t31 t32 t33 t41 t42 t43 
t51 t52 t53 t61 t62 t63 t71 t72 t73 t81 t82 t83);
v11 = b11 - (b11 + b12) / 3;
v12 = b12 - (b11 + b12) / 3;
v13 = - (b11 + b12) / 3;
v21 = b21 - (b21 + b22) / 3;
v22 = b22 - (b21 + b22) / 3;
v23 = - (b21 + b22) / 3;
v31 = b31 - (b31 + b32) / 3;
v32 = b32 - (b31 + b32) / 3;
v33 = - (b31 + b32) / 3;
v41 = b41 - (b41 + b42) / 3;
v42 = b42 - (b41 + b42) / 3;
v43 = - (b41 + b42) / 3;
v51 = b51 - (b51 + b52) / 3;
v52 = b52 - (b51 + b52) / 3;
v53 = - (b51 + b52) / 3;
v61 = b61 - (b61 + b62) / 3;
v62 = b62 - (b61 + b62) / 3;
v63 = - (b61 + b62) / 3;
v71 = b71 - (b71 + b72) / 3;
v72 = b72 - (b71 + b72) / 3;
v73 = - (b71 + b72) / 3;
v81 = b81 - (b81 + b82) / 3;
v82 = b82 - (b81 + b82) / 3;
v83 = - (b81 + b82) / 3;
t11 = a11 - (a11 + a12) / 3;
t12 = a12 - (a11 + a12) / 3;
t13 = - (a11 + a12) / 3;
t21 = a21 - (a21 + a22) / 3;
t22 = a22 - (a21 + a22) / 3;
t23 = - (a21 + a22) / 3;
t31 = a31 - (a31 + a32) / 3;
t32 = a32 - (a31 + a32) / 3;
t33 = - (a31 + a32) / 3;
t41 = a41 - (a41 + a42) / 3;
t42 = a42 - (a41 + a42) / 3;
t43 = - (a41 + a42) / 3;
t51 = a51 - (a51 + a52) / 3;
t52 = a52 - (a51 + a52) / 3;
t53 = - (a51 + a52) / 3;
t61 = a61 - (a61 + a62) / 3;
t62 = a62 - (a61 + a62) / 3;
t63 = - (a61 + a62) / 3;
t71 = a71 - (a71 + a72) / 3;
t72 = a72 - (a71 + a72) / 3;
t73 = - (a71 + a72) / 3;
t81 = a81 - (a81 + a82) / 3;
t82 = a82 - (a81 + a82) / 3;
t83 = - (a81 + a82) / 3;







Deviation contrasts can be estimated in flexMIRT using user defined contrasts matrices in the <Groups> section of the code. The commands Tc and Ta are used to specify a contrasts matrix for the intercepts and the slopes.

<Project>
Title = "nominal model fitted to assertiveness data";
Description = "8 items 1 Factor";

<Options>
Mode = Calibration;
M2 = Full;
GOF = Extended;
Processors = 4;
Quadrature = 49, 5.0;
Etol = 1e-2;
Mtol = 1e-2;
MaxE = 2000;
MaxM = 500;


<Groups>
%Group1%
 File = "assertivenessflexMIRT.dat";
 Varnames = v1-v8;
 N = 797;
 Ncats(v1-v8) = 3;
 Model(v1-v8) = Nominal(3);
Tc(v1-v8) = (
 1  0,
 0  1,
 -1 -1 );
Ta(v1-v8) = (
 1  0,
 0  1,
 -1 -1 );

<Constraints>



[bookmark: OLE_LINK8][bookmark: OLE_LINK9][bookmark: OLE_LINK10]Appendix D: Hypothesis testing on the estimated parameters for the assertiveness application


The following Mplus code estimates the difference between the two intercepts of the first item, , and provides a p-value for the hypothesis that the difference is zero. The difference is defined in the MODEL CONSTRAINT part of the code.

TITLE: Factor analysis of assertiveness data;
DATA:
    FILE IS assertiveness.dat;
          
VARIABLE:
    NAMES ARE i1-i8;
    USEVARIABLES ARE i1-i8;
!This sets the items to be nominal
    NOMINAL ARE i1-i8;

ANALYSIS:
    ESTIMATOR = ML;

MODEL:
   assertive BY i1#1*1;
   assertive BY i1#2*1;
   assertive BY i2#1*1;
   assertive BY i2#2*1;
   assertive BY i3#1*-1;
   assertive BY i3#2*1;
   assertive BY i4#1*1;
   assertive BY i4#2*1;
   assertive BY i5#1*-1;
   assertive BY i5#2*1;
   assertive BY i6#1*1;
   assertive BY i6#2*-1;
   assertive BY i7#1*-1;
   assertive BY i7#2*-1;
   assertive BY i8#1*-1;
   assertive BY i8#2*1;
   [i1#1] (a11);
   [i1#2] (a12);

   !Factor variance is set at 1.
    assertive@1;

MODEL CONSTRAINT: 
NEW (d1);
d1 = a11 - a12;



Appendix E. Scale of attitudes towards immigration

We provide below an English translation of the original items written in Spanish

1. What do you think should be done about immigrants that live in Spain illegally (no visa, expired visa, etc.)
· Regularize the status of all of them
· Regularize the status of those that have lived in Spain for several years, regardless of whether they are currently employed
· Regularize the status of those that are currently employed, regardless of how long they have been living in Spain.
· Maintain the current situation.
· Return them to their country of origin.

2. What would you do about immigrants that apply for asylum?
· Grant asylum with no restrictions.
· Grant asylum only if they can prove they are being prosecuted in their country of origin
· Grant asylum only to a limited number of applicants who are being prosecuted in their country of origin
· Deny asylum to all applicants

3. Do you think the immigration laws in Spain are
· Too tolerant.
· Tolerant.
· Fair.
· A little bit strict.
· Too strict.

4. How do you thing that Spaniards treat immigrants.
· With indifference.
· With contempt.
· With kindness.
· With mistruth.
· Normally, as if they were Spaniards.
· Aggressively.

5. What is the main factor in how immigrants are treated in Spain? The immigrants
· Nationality.
· Culture.
· Economic status.
· Skin color.
· Religion.
· Ethnic group.

6. How do you thing that immigrants treat the Spanish people.
· With indifference.
· With contempt.
· With kindness.
· With mistruth.
· With normality.
· Aggressively.
Appendix F: Mplus code for the scale of attitudes toward immigration

The following Mplus code estimates the three-factor model.

  TITLE: Thissen, Cai and Bock model
  DATA:  FILE IS immigration.dat;
          
  VARIABLE: NAMES ARE v1-v6;
          USEVARIABLES ARE v1-v6;
          NOMINAL ARE v1-v6;
          MISSING ARE ALL (0);

  ANALYSIS: ESTIMATOR = ML;

  MODEL:  FACTOR1 BY v1#1* ;
          FACTOR1 BY v1#2* ;
          FACTOR1 BY v1#3* ;
          FACTOR1 BY v1#4* ;

          FACTOR1 BY v2#1* ; 
          FACTOR1 BY v2#2* ;
          FACTOR1 BY v2#3* ;

          FACTOR1 BY v3#1* ; 
          FACTOR1 BY v3#2* ;
          FACTOR1 BY v3#3* ;
          FACTOR1 BY v3#4* ;

          FACTOR1 BY v4#1* ; 
          FACTOR1 BY v4#2* ;
          FACTOR1 BY v4#3* ;
          FACTOR1 BY v4#4* ;
          FACTOR1 BY v4#5* ;

          FACTOR1 BY v5#1* ; 
          FACTOR1 BY v5#2* ;
          FACTOR1 BY v5#3* ;
          FACTOR1 BY v5#4* ;
          FACTOR1 BY v5#5* ;

          FACTOR1 BY v6#1* ; 
          FACTOR1 BY v6#2* ;
          FACTOR1 BY v6#3* ;
          FACTOR1 BY v6#4* ;
          FACTOR1 BY v6#5* ;

          FACTOR2 BY v1#1@0 ; ! loading set to zero to fix rotation
          FACTOR2 BY v1#2* ;
          FACTOR2 BY v1#3* ;
          FACTOR2 BY v1#4* ;

          FACTOR2 BY v2#1* ; 
          FACTOR2 BY v2#2* ;
          FACTOR2 BY v2#3* ;

          FACTOR2 BY v3#1* ; 
          FACTOR2 BY v3#2* ;
          FACTOR2 BY v3#3* ;
          FACTOR2 BY v3#4* ;

          FACTOR2 BY v4#1* ; 
          FACTOR2 BY v4#2* ;
          FACTOR2 BY v4#3* ;
          FACTOR2 BY v4#4* ;
          FACTOR2 BY v4#5* ;

          FACTOR2 BY v5#1* ; 
          FACTOR2 BY v5#2* ;
          FACTOR2 BY v5#3* ;
          FACTOR2 BY v5#4* ;
          FACTOR2 BY v5#5* ;

          FACTOR2 BY v6#1* ; 
          FACTOR2 BY v6#2* ;
          FACTOR2 BY v6#3* ;
          FACTOR2 BY v6#4* ;
          FACTOR2 BY v6#5* ;

          FACTOR3 BY v1#1@0 ; ! loading set to zero to fix rotation
          FACTOR3 BY v1#2@0 ; ! loading set to zero to fix rotation
          FACTOR3 BY v1#3* ;
          FACTOR3 BY v1#4* ;

          FACTOR3 BY v2#1* ; 
          FACTOR3 BY v2#2* ;
          FACTOR3 BY v2#3* ;

          FACTOR3 BY v3#1* ; 
          FACTOR3 BY v3#2* ;
          FACTOR3 BY v3#3* ;
          FACTOR3 BY v3#4* ;

          FACTOR3 BY v4#1* ; 
          FACTOR3 BY v4#2* ;
          FACTOR3 BY v4#3* ;
          FACTOR3 BY v4#4* ;
          FACTOR3 BY v4#5* ;

          FACTOR3 BY v5#1* ; 
          FACTOR3 BY v5#2* ;
          FACTOR3 BY v5#3* ;
          FACTOR3 BY v5#4* ;
          FACTOR3 BY v5#5* ;

          FACTOR3 BY v6#1* ; 
          FACTOR3 BY v6#2* ;
          FACTOR3 BY v6#3* ;
          FACTOR3 BY v6#4* ;
          FACTOR3 BY v6#5* ;

         !Standardize factors and set factors uncorrelated
          FACTOR1@1;
          FACTOR2@1;
          FACTOR3@1;
          FACTOR1 with FACTOR2@0;
          FACTOR1 with FACTOR3@0;
          FACTOR2 with FACTOR3@0;

Appendix G: Data analysis examination
Below we provide the Data Analysis exam items. The correct option is boldfaced.

Exercise 1. A social researcher has conducted a study to determine if individuals’ educational level (measured as primary, secondary, medium, and superior studies) is related to the extent of their worry over the current economic crisis (measured as low, medium, and high).

Please answer the questions below.

1. These variables are…
(a) Both categorical
(b) Both quantitative
(c) One is categorical, the other is quantitative

2. The null hypothesis (H0) tested in this study is… 
(a) Individuals with lower educational levels are more worried over the economic crisis
(b) Individuals with higher educational levels are more worried over the economic crisis
(c) Worry over the economic crisis is independent of educational level

3. To test this null hypothesis, the researcher has used the X2 Pearson statistic. The table below offers some quantiles of the sample distribution of the X2 statistic under H0 and H1:

	X2
	0
	1.64
	2.20
	10.64
	12.59
	16.81
	18.55

	F(X2|H0)
	.00
	.05
	.10
	.90
	.95
	.99
	.995

	F(X2|H1)
	.00
	.01
	.05
	.65
	.75
	.90
	.95



Assuming  = .05, the critical value is…
(a) 1.64
(b) 12.59
(c) 18.55

4. If the X2 statistic takes a value of 10.64 in a specific sample, then we conclude  that…
(a) The degree of worry over the economic crisis is independent of educational level
(b) The degree of worry over the economic crisis is related to educational level
(c) There is no evidence that supports that the degree of worry over the economic crisis is related to educational level.

5. What is the value of statistical power of the test?
(a) .25
(b) .75
(c) .95

Exercise 2. We wish to investigate whether males and females differ in their agreement with current government policies. A random sample of 200 participants is obtained, half of them being males. We obtain their opinion on a scale from 0 to 100 points (0 = completely disagree, 100 = completely agree).

Please answer the questions below.

6. What variables are used in the study?
(a) Opinion of the respondents
(b) Gender and opinion of the respondents
(c) Gender, opinion of the respondents, and decisions of politicians

7. What statistical test can be used to compare the opinion of males and females?
(a) The X2 Pearson test of independence
(b) The t- test for independent samples
(c) A two-way ANOVA with gender and decisions of politicians as independent variables and opinion of the respondents as dependent variable

8. If the test statistic takes the value 5, then, should we reject the null hypothesis?
(a) Yes
(b) No
(c) There is not enough information to make a decision about the null hypothesis

[bookmark: OLE_LINK11][bookmark: OLE_LINK12]

Appendix H: Mplus code for the Data Analysis examination

TITLE: Bifactor model with uncorrelated factors: Data Analysis exam;
DATA:
    FILE IS examination.dat;
 
VARIABLE:
    NAMES ARE i1-i10;
    USEVARIABLES ARE i1-i10;
    NOMINAL ARE i1-i10;

ANALYSIS:
    ESTIMATOR = ML; 

MODEL:
   g BY i1#1*-1;
   g BY i1#2*-1;
   g BY i2#1*-1;
   g BY i2#2*-1;
   g BY i3#1*-1;
   g BY i3#2*-1;
   g BY i4#1*-1;
   g BY i4#2*-1;
   g BY i5#1*-1;
   g BY i5#2*-1;
   g BY i6#1*-1;
   g BY i6#2*-1;
   g BY i7#1*-1;
   g BY i7#2*-1;
   g BY i8#1*-1;
   g BY i8#2*-1;

   f1 BY i1#1*-1;
   f1 BY i1#2*-1;
   f1 BY i2#1*-1;
   f1 BY i2#2*-1;
   f1 BY i3#1*-1;
   f1 BY i3#2*-1;
   f1 BY i4#1*-1;
   f1 BY i4#2*-1;
   f1 BY i5#1*-1;
   f1 BY i5#2*-1;

   f2 BY i6#1*-1;
   f2 BY i6#2*-1;
   f2 BY i7#1*-1;
   f2 BY i7#2*-1;
   f2 BY i8#1*-1;
   f2 BY i8#2*-1;
 
   [ i1#1*1];
   [ i1#2*1];
   [ i2#1*1];
   [ i2#2*1];
   [ i3#1*1];
   [ i3#2*1];
   [ i4#1*1];
   [ i4#2*1];
   [ i5#1*1];
   [ i5#2*1];
   [ i6#1*1];
   [ i6#2*1];
   [ i7#1*1];
   [ i7#2*1];
   [ i8#1*1];
   [ i8#2*1];
   [ i9#1*1];
   [ i9#2*1];
   [ i10#1*1];
   [ i10#2*1];
     
! factor variances are set to 1 for identification
	g-f2@1;
! general factor uncorrelated with group factors for identification
	g WITH f1-f2@0;
! group factors uncorrelated 
    f1 WITH f2@0;




Appendix I: Items to measure emotional stability


	Item
	Item stem

	1
	I am so jittery, that I cannot bear  certain sounds (e.g., the squeak of a door)

	2
	I think that I am more nervous than most people

	3
	My muscles tend to be in tension

	4
	Sometimes I am in such a bad mood that I feel like breaking things

	5
	Very small setbacks irritate me a lot

	6
	I go nuts over small things, although I realize they are trivial.

	7
	The fiercest fights are the ones that I have with myself.

	8
	I have memories or thoughts that are constantly spinning in my head.

	9
	I am still seriously worried about the mistakes I made in the past.

	10
	My emotions are so illogical, that I am unable to control them



Items are scored using a four-point rating scale: Completely Disagree, Disagree, Agree and Completely Agree.



Appendix K: flexMIRT code for estimating Emotional Stability

The following code estimates the full rank model. Identity-based contrast matrix have been used for numerical stability during estimation. The line of code Code(v1-v10) = (1,2,3,4),(3,2,1,0); means that in the data file the categories are scored in reverse order and from 1 to 4, and the read data are reversed and scored from 0 to 3 prior to analysis.


<Project>
Title = "nominal model fitted to the emotional stability scale";
Description = "10 items 1 Factor. Full rank model";

<Options>
Mode = Calibration;
M2 = Full;
GOF = Extended;
Processors = 2;
Quadrature = 49, 4.0;
Etol = 1e-2;
Mtol = 1e-2;
MaxE = 2000;
MaxM = 500;

<Groups>
%Group1%
 File = "stability.dat";
 Varnames = v1-v10;
 N = 1000;

 Ncats(v1-v10) = 4;
 Code(v1-v10) = (1,2,3,4),(3,2,1,0);
 Model(v1-v10) = Nominal(4);

 Ta(v1-v10)=Identity;
 Tc(v1-v10)=Identity;

<Constraints>







The following code estimates the model with . The constraint is imposed by assuming a Fourier-based contrast matrix for the slopes (Ta(v1-v10)=Trend;) and fixing the third scoring function to its zero default value (Fix(v1-v10),ScoringFn(3);).

<Project>
Title = "nominal model fitted to the emotiona stability scale";
Description = "10 items 1 Factor. 1 fixed scoring basic parameter";

<Options>
Mode = Calibration;
M2 = Full;
GOF = Extended;
Processors = 2;
Quadrature = 49, 4.0;
Etol = 1e-2;
Mtol = 1e-2;
MaxE = 2000;
MaxM = 500;

<Groups>
%Group1%
 File = "stability.dat";
 Varnames = v1-v10;
 N = 1000;

 Ncats(v1-v10) = 4;
 Code(v1-v10) = (1,2,3,4),(3,2,1,0);
 Model(v1-v10) = Nominal(4);

 Ta(v1-v10)=Trend;
 Tc(v1-v10)=Identity;

<Constraints>
Fix(v1-v10),ScoringFn(3);







The flexMIRT code for the model with , equivalent to a generalized partial credit model, is


<Project>
Title = "nominal model fitted to the emotiona stability scale";
Description = "10 items 1 Factor. 2 fixed scoring basic parameters";

<Options>
Mode = Calibration;
M2 = Full;
GOF = Extended;
Processors = 2;
Quadrature = 49, 4.0;
Etol = 1e-2;
Mtol = 1e-2;
MaxE = 2000;
MaxM = 500;


<Groups>
%Group1%
 File = "stability.dat";
 Varnames = v1-v10;
 N = 1000;

 Ncats(v1-v10) = 4;
 Code(v1-v10) = (1,2,3,4),(3,2,1,0);
 Model(v1-v10) = Nominal(4);

 Ta(v1-v10)=Trend;
 Tc(v1-v10)=Identity;

<Constraints>
Fix(v1-v10),ScoringFn(2);
Fix(v1-v10),ScoringFn(3);



Appendix K: Mplus code for estimating Emotional Stability

The following Mplus code estimates Model 3 in the example of emotional stability. Factor 1 is Emotional Stability and Factor 2 is ERS.


  TITLE: Thissen, Cai and Bock model
  DATA:
          FILE IS stability.dat;
          
  VARIABLE:
          NAMES ARE v1-v10;
          USEVARIABLES ARE v1-v10;
          NOMINAL ARE v1-v10;
          MISSING ARE ALL (0);

  ANALYSIS:
          ESTIMATOR = ML;
          INTEGRATION = GAUSSHERMITE(15); ! # of quadrature points
  MODEL:
          THETA1 BY v1#1* (L111);
          THETA1 BY v1#2* (L121);
          THETA1 BY v1#3* (L131);

          THETA1 BY v2#1* (L211); 
          THETA1 BY v2#2* (L221);
          THETA1 BY v2#3* (L231);

          THETA1 BY v3#1* (L311); 
          THETA1 BY v3#2* (L321);
          THETA1 BY v3#3* (L331);

          THETA1 BY v4#1* (L411); 
          THETA1 BY v4#2* (L421);
          THETA1 BY v4#3* (L431);

          THETA1 BY v5#1* (L511); 
          THETA1 BY v5#2* (L521);
          THETA1 BY v5#3* (L531);

          THETA1 BY v6#1* (L611); 
          THETA1 BY v6#2* (L621);
          THETA1 BY v6#3* (L631);

          THETA1 BY v7#1* (L711); 
          THETA1 BY v7#2* (L721);
          THETA1 BY v7#3* (L731);

          THETA1 BY v8#1* (L811); 
          THETA1 BY v8#2* (L821);
          THETA1 BY v8#3* (L831);

          THETA1 BY v9#1* (L911); 
          THETA1 BY v9#2* (L921);
          THETA1 BY v9#3* (L931);

          THETA1 BY v10#1* (L1011); 
          THETA1 BY v10#2* (L1021);
          THETA1 BY v10#3* (L1031);

          THETA2 BY v1#2* (L122);
          THETA2 BY v1#3* (L132);

          THETA2 BY v2#2* (L222);
          THETA2 BY v2#3* (L232);

          THETA2 BY v3#2* (L322);
          THETA2 BY v3#3* (L332);

          THETA2 BY v4#2* (L422);
          THETA2 BY v4#3* (L432);

          THETA2 BY v5#2* (L522);
          THETA2 BY v5#3* (L532);

          THETA2 BY v6#2* (L622);
          THETA2 BY v6#3* (L632);

          THETA2 BY v7#2* (L722);
          THETA2 BY v7#3* (L732);

          THETA2 BY v8#2* (L822);
          THETA2 BY v8#3* (L832);

          THETA2 BY v9#2* (L922);
          THETA2 BY v9#3* (L932);

          THETA2 BY v10#2* (L1022);
          THETA2 BY v10#3* (L1032);

          [v1#1];
          [v1#2];
          [v1#3];
 
          [v2#1];
          [v2#2];
          [v2#3];
 
          [v3#1];
          [v3#2];
          [v3#3];
 
          [v4#1];
          [v4#2];
          [v4#3];
 
          [v5#1];
          [v5#2];
          [v5#3];
 
          [v6#1];
          [v6#2];
          [v6#3];
 
          [v7#1];
          [v7#2];
          [v7#3];
 
          [v8#1];
          [v8#2];
          [v8#3];
 
          [v9#1];
          [v9#2];
          [v9#3];
 
          [v10#1];
          [v10#2];
          [v10#3];
 
         !Standardize factors
          THETA1@1;
          THETA2@1;

  MODEL CONSTRAINT:
          !Implements constraints implied by the TCB model
           NEW(a1_1); NEW(a2_1); NEW(a3_1); NEW(a4_1); NEW(a5_1);
           NEW(a6_1); NEW(a7_1); NEW(a8_1); NEW(a9_1); NEW(a10_1);
           NEW(a1_2); NEW(a2_2); NEW(a3_2); NEW(a4_2); NEW(a5_2);
           NEW(a6_2); NEW(a7_2); NEW(a8_2); NEW(a9_2); NEW(a10_2);
           NEW(alpha11); NEW(alpha12); NEW(alpha21); NEW(alpha22); 
           NEW(alpha31); NEW(alpha32); NEW(alpha41); NEW(alpha42); 
           NEW(alpha51); NEW(alpha52); NEW(alpha61); NEW(alpha62); 
           NEW(alpha71); NEW(alpha72); NEW(alpha81); NEW(alpha82); 
           NEW(alpha91); NEW(alpha92); NEW(alpha101); NEW(alpha102); 

            L131 = a1_1*(1 + 0.866*alpha11 + 0.866*alpha12);
            L121 = a1_1*(2 + 0.866*alpha11 - 0.866*alpha12);
            L111 = a1_1*3;
            L231 = a2_1*(1 + 0.866*alpha21 + 0.866*alpha22);
            L221 = a2_1*(2 + 0.866*alpha21 - 0.866*alpha22);
            L211 = a2_1*3;
            L331 = a3_1*(1 + 0.866*alpha31 + 0.866*alpha32);
            L321 = a3_1*(2 + 0.866*alpha31 - 0.866*alpha32);
            L311 = a3_1*3;
            L431 = a4_1*(1 + 0.866*alpha41 + 0.866*alpha42);
            L421 = a4_1*(2 + 0.866*alpha41 - 0.866*alpha42);
            L411 = a4_1*3;
            L531 = a5_1*(1 + 0.866*alpha51 + 0.866*alpha52);
            L521 = a5_1*(2 + 0.866*alpha51 - 0.866*alpha52);
            L511 = a5_1*3;
            L631 = a6_1*(1 + 0.866*alpha61 + 0.866*alpha62);
            L621 = a6_1*(2 + 0.866*alpha61 - 0.866*alpha62);
            L611 = a6_1*3;
            L731 = a7_1*(1 + 0.866*alpha71 + 0.866*alpha72);
            L721 = a7_1*(2 + 0.866*alpha71 - 0.866*alpha72);
            L711 = a7_1*3;
            L831 = a8_1*(1 + 0.866*alpha81 + 0.866*alpha82);
            L821 = a8_1*(2 + 0.866*alpha81 - 0.866*alpha82);
            L811 = a8_1*3;
            L931 = a9_1*(1 + 0.866*alpha91 + 0.866*alpha92);
            L921 = a9_1*(2 + 0.866*alpha91 - 0.866*alpha92);
            L911 = a9_1*3;
            L1031 = a10_1*(1 + 0.866*alpha101 + 0.866*alpha102);
            L1021 = a10_1*(2 + 0.866*alpha101 - 0.866*alpha102);
            L1011 = a10_1*3;

            L122 = -a1_2;
            L132 = -a1_2;
            L222 = -a2_2;
            L232 = -a2_2;
            L322 = -a3_2;
            L332 = -a3_2;
            L422 = -a4_2;
            L432 = -a4_2;
            L522 = -a5_2;
            L532 = -a5_2;
            L622 = -a6_2;
            L632 = -a6_2;
            L722 = -a7_2;
            L732 = -a7_2;
            L822 = -a8_2;
            L832 = -a8_2;
            L922 = -a9_2;
            L932 = -a9_2;
            L1022 = -a10_2;
            L1032 = -a10_2;

Estimation of the TCB parameterization is much more straightforward in flexMIRT because the Fourier based and Identity contrast matrices are already build-up in the computer program. The user has only to indicate which contrast matrix to use. In the example, the flexMIRT codes Ta(v1-v10)=Trend; and Tc(v1-v10)=Identity;  included in the <Groups> section specify a Fourier-based T matrix for the slopes and an identity based T matrix for the intercepts. The ordering of the categories is estimated by comparing models with different levels of flexibility in the a1 and a2 parameters in Eq. (15). The full rank model assumes that both a1 and a2 are free parameters. The second model assumes that a2 = 0. This is achieved by including the command Fix(v1-v10),ScoringFn(3); in the <Constraints> section of flexMIRT code (the token ScoringFn(3)refers to a2 because in flexMIRT ScoringFn(1) is the constant value 1 in Eq. (15), and  ScoringFn(2) is a1). The third model is the generalized partial credit model which assumes a1 = a2 = 0, and is specified by the commands Fix(v1-v10), ScoringFn(2); and Fix(v1-v10),ScoringFn(3);.

